Geodesic Motion in the Schwarzschild Metric

The Geodesic Equations for Massive Particles

The geodesic motion of freely falling particles can be obtained from the solution of the geodesic equations. We choose the coordinates so that the motion is confined in the $\theta = \frac{\pi}{2}$ plane. Therefore $\dot{\theta} = 0,\ \sin(\theta) = 1$. Then we have to solve the system

$$\ddot{r} = \frac{\dot{r}^2 - e^2}{r^2\left(1 - \frac{2}{r}\right)} + \frac{l^2}{r^3}\left(1 - \frac{2}{r}\right)$$

$$\dot{t} = \frac{e}{1 - \frac{2}{r}}$$

$$\dot{\phi} = \frac{l}{r^2}$$

where $\dot{t} = \frac{d}{d\tau}\, t(\tau)$ etc, and the conserved quantities

$$e = \left(1 - \frac{2}{r}\right)\dot{t}$$

$$l = r^2\, \dot{\phi}$$

are given. The radial motion "conservation of energy" is given by

$$\mathcal{E} = \frac{e^2 - 1}{2} = \frac{1}{2}\, \dot{r}^2 + V_{\text{eff}}(r)$$

$$V_{\text{eff}}(r) = \frac{1}{2}\left[\left(1 - \frac{2}{r}\right)\left(1 + \frac{l^2}{r^2}\right) - 1\right] = -\frac{1}{r} + \frac{l^2}{2\, r^2} - \frac{l^2}{r^3}$$

The turning points of the radial motion are calculated from the equation (type $\mathcal{E}$ by [esc]scE[esc])

$$V_{\text{eff}}(r) = \mathcal{E}$$

In the above equations, dimensionless quantities are used. To go back to geometrized units, use the dictionary:

$$\tau \to \frac{\tau}{M} \qquad t \to \frac{t}{M} \qquad r \to \frac{r}{M} \qquad \phi \to \phi$$

$$l \to \frac{l}{M} \qquad \mathcal{E} \to \mathcal{E} \qquad V_{\text{eff}} \to V_{\text{eff}} \qquad e \to e$$

Effective potential for radial motion of massive particle

Plot $V_{\text{eff}}(r)$ for various angular momenta

In[ ]:=
```
l =.;
Veff[r_] := -1/r + l^2/(2 r^2) - l^2/r^3;
Plot[Evaluate[{
   Veff[r] /. l → 3.2,
   Veff[r] /. l → 4.0,
   Veff[r] /. l → 4.6,
   Veff[r] /. l → 5.0
  }], {r, 2, 80}, PlotRange → {-0.075, 0.1},
 PlotLegends → {"3.2", "4.0", "4.6", "5.0"}, ImageSize → Large]
```

Out[ ]=



Compute turning points:

l is l/M

```
l = 4.3; rmax = 40.;
Veff[r_] := -1/r + l^2/(2 r^2) - l^2/r^3;
Plot[Evaluate[{
   Veff[r],
   Veff[rmax]
  }], {r, 2, 80},
 PlotRange    → {-0.05, 0.05},
 PlotLegends  → {"l=" <> ToString[l], "E= " <> ToString[Veff[rmax]]},
 PlotLabel    →
  "Turning Points: " <> ToString[r /. NSolve[Veff[r] == Veff[rmax], r]],
 ImageSize    → Large]
```

Out[ ]=



Geodesic Equations

In[ ]:=
```
Veff[r_] := -1/r + l^2/(2 r^2) - l^2/r^3;

solveGeodesicEqs[e_, l_, ϕ0_, r0_, vr0_, τfinal_] := NDSolve[{
    t'[τ]  == e/(1 - 2/r[τ]),
    ϕ'[τ]  == l/r[τ]^2 ,
    r''[τ] == -e^2/(r[τ]^2 (1 - 2/r[τ])) + (r'[τ])^2/(r[τ]^2 (1 - 2/r[τ])) + (1 - 2/r[τ]) l^2/r[τ]^3,
    t[0] == 0, ϕ[0] == ϕ0, r[0] == r0,
    r'[0] == vr0  (* initial conditions                 *)
  }, {t, ϕ, r}, {τ, 0, τfinal}
 ];
```

Plotting function

Skip this section if you are not interested in the plotting commands.
This is the function that plots the results.

```
eRangeDef = {{0, 40}, {-0.04, 0.05}};
(* use this variable for eRange below for most plots *)
plotResults[xyRange_, eRange_] :=
 (* xyRange is the range of the x-y   plot           *)
 Module[  {xyrange = xyRange},
  (* eRange  is the range in the ℰ/V_eff-plot             *)
  GraphicsGrid[{                              (* Arrange
     the plots in a grid                           *)
    { (* 1st row *)
     Show[{
       ParametricPlot[Evaluate[
         {r[τ] Cos[ϕ[τ]], r[τ] Sin[ϕ[τ]]}
          (* x=r cos(ϕ) y=r sin(ϕ)                            *)
          /. sol], {τ, 0.1, τmax}, AxesLabel → {"x", "y"}, PlotRange → xyrange],
       Graphics[{Red, Disk[{0, 0}, 2]}]
```

```
      (* a red disk in the black hole area                   *)
     }],
    Plot[{
      Veff[r], Energy
      (* The effective potential with the energy level E    *)
     }, {r, 0, 40},
     PlotRange → eRange,
     PlotLabel → "Turning Points: "     (* compute
        the turning points                              *)
       <> ToString[Select[r /. NSolve[Veff[r] == Energy , r], # ∈ Reals && # > 0 &]]
       <> "     ℰ = " <> ToString[Energy],
     (*  display the value of E on the plot              *)

     AxesLabel → {"r", "V_eff[r]"}]
   }, (* 1st row *)
   {  (* 2nd row *)
    Plot[Evaluate[{
        r[τ]
        (* radial  coordinate as a function of λ             *)
      } /. sol], {τ, 0, τmax}, AxesLabel → {"τ", "r[τ]"}, PlotRange → All],
    Plot[Evaluate[{
        {Mod[ϕ[τ], 2 π], 2 π}
        (* angular coordinate as a function of λ             *)
      } /. sol], {τ, 0, τmax}, AxesLabel → {"τ", "ϕ[τ]"}, PlotRange → All]
   }, (* 2nd row *)
   {  (* 3rd row *)
    Plot[Evaluate[{
        t[τ] / τ
        (* time  coordinate as a function of λ             *)
      } /. sol], {τ, 0, τmax}, AxesLabel → {"τ", "γ[τ]≡ t[τ]/τ "}, PlotRange → All],
    Plot[Evaluate[
      (* check numerical errors in ℰ                         *)
      (0.5 r'[τ]^2 + Veff[r[τ]] - Energy) /. sol
     ], {τ, 0, τmax}, AxesLabel → {"τ", "Δℰ"}, PlotLabel → "Error in ℰ"]
   } (* 3rd row *)

  }, ImageSize → Full]
]
```

Solutions to the geodesic equations for massive particles

Set:

r1: one of the turning points

l : the angular momentum

r0: initial radial position

τmax: maximum time for numerical integration

radialdirection: the sign of the initial radial velocity $\dot{r}(0)$

r1 will determine the energy $\mathcal{E}$ and the conserved quantity e, which in turn will determine $|\dot{r}(0)|$

v0 = (radialdirection) × $|\dot{r}(0)|$

Bound trajectories

In[ ]:=
```
r1   = 30.0  ; l = 4.3;  τmax = 10 300 ;
r0   = 29.5  ; radialdirection = -1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Veff[r1]; e = √(1 + 2 Energy) ;
v0 = radialdirection √(2 (Energy - Veff[r0])) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=


Turning Points: {30., 9.33081, 2.78168}   ℰ = −0.023745
y
x
$V_{eff}[r]$
r
r[τ]
τ
ϕ[τ]
τ
$\gamma[\tau]\equiv\frac{t[\tau]}{\tau}$
τ
Error in ℰ
Δℰ
τ
$-5.\times10^{-8}$
$-1.\times10^{-7}$
-1.5×10^{-7}
$-2.\times10^{-7}$

In[ ]:=
```
r1 = 500.0; l = 4.3;
τmax = 132 000;
r0 = 490.5; radialdirection = -1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Veff[r1]; e = Sqrt[1 + 2 Energy];
v0 = radialdirection Sqrt[2 (Energy - Veff[r0])];
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=



Scattering

Set:

b: the impact parameter

x0: initial x coordinate

V0: v_0, the x-component of the velocity



Then:

$r_0 \quad = \sqrt{{x_0}^2 + b^2}$

$\dot{r}_0 \quad = v_0 \cos(\phi_0) \qquad$ We assume $(r \gg 2)$: $v^r = \frac{\mathrm{d}r}{\mathrm{d}t} = \frac{\mathrm{d}r}{\mathrm{d}\tau}\frac{\mathrm{d}\tau}{\mathrm{d}t} \approx \frac{\mathrm{d}r}{\mathrm{d}\tau} = \dot{r}$

$r_0\ \dot{\phi}_0 = v_0 \sin(\phi_0) \qquad \frac{\mathrm{d}\phi}{\mathrm{d}t} \approx \frac{\mathrm{d}\phi}{\mathrm{d}\tau}$

$l \quad = {r_0}^2\ \dot{\phi}_0 \qquad = r_0\, v_0 \sin(\phi_0)$

$\mathcal{E} \quad = \frac{1}{2}\,{\dot{r}_0}^2 + V_{\mathrm{eff}}(r_0)$

In[ ]:=
```
b = 15.; x0 = 500.; V0 = 0.30;
τmax = 1800;
r0 = Sqrt[x0^2 + b^2];
ϕ0 = ArcTan[b/x0];
v0 =   -V0 Cos[ϕ0];
l  =  r0 V0 Sin[ϕ0];

Energy = 0.5 v0^2 + Veff[r0]; e = Sqrt[1 + 2 Energy];

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[{{-10, 100}, {-50, 50}}, Automatic];

Print[
  "Scattering Angle: (degrees)\n" ×
   "θ= tan^-1 (v^y (∞))/(v^x (∞))  = ",
  180/π (ArcTan[(* ArcTan[x,y] gives angle in the range (-π,π]*)
      r '[τmax] Cos[ϕ[τmax]] - ϕ '[τmax] r[τmax] Sin[ϕ[τmax]] /. sol,
      r '[τmax] Sin[ϕ[τmax]] + ϕ '[τmax] r[τmax] Cos[ϕ[τmax]] /. sol
    ])
 ];
Show[gg]
```

```
Scattering Angle: (degrees)
θ= tan^-1 (v^y (∞))/(v^x (∞))  = {70.6345}
```

Out[ ]=


Turning Points: {4.99997, 3.01222}   ℰ = 0.0430007
y
x
V_{eff}[r]
r
r[τ]
τ
ϕ[τ]
τ
γ[τ]≡t[τ]/τ
τ
Error in ℰ
Δℰ
τ

In[ ]:=
```
b = 15.; x0 = 500.; V0 = 0.30;
τmax = 1800;
r0 = Sqrt[x0^2 + b^2];
ϕ0 = ArcTan[b/x0];
v0 =   -V0 Cos[ϕ0];
l  = r0 V0 Sin[ϕ0];

Energy = 0.5 v0^2 + Veff[r0]; e = Sqrt[1 + 2 Energy];

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[{{-10, 500}, {-10, 30}}, Automatic];

Print[
  "Scattering Angle: (degrees)\n" ×
   "θ= tan^-1 (v^y (∞))/(v^x (∞))  = ",
  180/π (ArcTan[(* ArcTan[x,y] gives angle in the range (-π,π]*)
      r '[τmax] Cos[ϕ[τmax]] - ϕ '[τmax] r[τmax] Sin[ϕ[τmax]] /. sol,
      r '[τmax] Sin[ϕ[τmax]] + ϕ '[τmax] r[τmax] Cos[ϕ[τmax]] /. sol
    ])
 ];
Show[gg]
```

Scattering Angle: (degrees)

$\theta = \tan^{-1}\frac{v^{y}(\infty)}{v^{x}(\infty)}$ = {70.6345}

Out[ ]=



Motion starting from $r < r_{max}$

Here a trajectory that cannot escape: $\mathcal{E} < V_{eff}(r_{max})$

We try to move outwards: $\dot{r}(0) > 0$

In[ ]:=

```
r1 = 5; l = 4.3;
τmax = 6.12;
r0 = 3; radialdirection = +1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Veff[r1]; e = √(1 + 2 Energy) ;
v0 = radialdirection √(2 (Energy - Veff[r0])) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=



Now we try to escape: Set $\mathcal{E} > V_{\text{eff}}(r_{\text{max}})$

In[ ]:=

```
Energy = 0.10; l = 4.3;
τmax  = 100.4;
r0    = 3; radialdirection = +1; (* Set it to +/- 1. The sign of ṙ(0)  *)

e  = √(1 + 2 Energy) ;
v0 = radialdirection √(2 (Energy - Veff[r0])) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=



But if we make the mistake to start with $\dot{r}(0) < 0$, then we fall into the BH

In[ ]:=

```
Energy = 0.10; l = 4.3;
τmax = 1.58;
r0 = 3; radialdirection = -1;(* Set it to +/- 1. The sign of ṙ(0)  *)

e = √(1 + 2 Energy);
v0 = radialdirection √(2 (Energy - Veff[r0]));
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=



Precession of the perihelion

In[ ]:=
```
r1 = 500.0; l = 4.3;
τmax = 132 000;
r0 = 490.5; radialdirection = -1; (* Set it to +/- 1. The sign of ṙ(0)  *)

Energy = Veff[r1]; e = √(1 + 2 Energy) ;
v0 = radialdirection √(2 (Energy - Veff[r0])) ;
ϕ0 = 0;

sol = solveGeodesicEqs[e, l, ϕ0, r0, v0, τmax];
gg = plotResults[All, eRangeDef]
```

Out[ ]=



Compute the precession angle of the motion:

In[ ]:=

```
(*Define ordinary functions for r and ϕ,with capital letters: *)
Φ[τ_] := ϕ[τ] /. sol⟦1, 2⟧;
R[τ_] := r[τ] /. sol⟦1, 3⟧;

(* from the plots above determine
 approximate τ near maxima and use them below: *)
τ1 = τ /. Last[FindMaximum[R[τ], {τ, 25 000}]];
τ2 = τ /. Last[FindMaximum[R[τ], {τ, 50 000}]];
τ3 = τ /. Last[FindMaximum[R[τ], {τ, 70 000}]];
τ4 = τ /. Last[FindMaximum[R[τ], {τ, 90 000}]];

(* compute the corresponding angles: *)
ϕ1 = Φ[τ1];
ϕ2 = Φ[τ2];
ϕ3 = Φ[τ3];
ϕ4 = Φ[τ4];

Print[
 "τ1= ", τ1, "  τ2= ", τ2, " τ3= ", τ3, " τ4= ", τ4, "\n",
 "ϕ1= ", ϕ1, " = ", Mod[ϕ1, 2 π] 180/π, "°      ",
 "ϕ2= ", ϕ2, " = ", Mod[ϕ2, 2 π] 180/π, "°      ",
 "ϕ3= ", ϕ3, " = ", Mod[ϕ3, 2 π] 180/π, "°      ",
 "ϕ4= ", ϕ4, " = ", Mod[ϕ4, 2 π] 180/π, "°      \n",
 "Define: δϕi= ϕi+1-ϕi-2π\n",
 "δϕ1= ", ϕ2 - ϕ1 - 2 π, " = ", (ϕ2 - ϕ1 - 2 π) 180/π, "°     ",
 "δϕ2= ", ϕ3 - ϕ2 - 2 π, " = ", (ϕ3 - ϕ2 - 2 π) 180/π, "°     ",
 "δϕ3= ", ϕ4 - ϕ3 - 2 π, " = ", (ϕ4 - ϕ3 - 2 π) 180/π, "°     "
]
```

FindMaximum: The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient increase in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

FindMaximum: The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient increase in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances.

τ_1= 23332. τ_2= 48877.3 τ_3= 74422.2 τ_4= 99966.9
ϕ_1= 8.73054 = 140.223° ϕ_2= 17.4997 = 282.656
° ϕ_3= 26.2688 = 65.089° ϕ_4= 35.0379 = 207.522°
Define: $\delta\phi_i = \phi_{i+1} - \phi_i - 2\pi$
$\delta\phi_1$= 2.48592 = 142.433° $\delta\phi_2$= 2.48592 = 142.433° $\delta\phi_3$= 2.48592 = 142.433°

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.
Email: konstant@mail.ntua.gr
Web: http://physics.ntua.gr/konstant